Solution of Exercises Sheet #6

Question 1:

Using a spreadsheet and the Monte Carlo method estimate the following integral with 99% confidence to within ±0.01?

$$\int_{0}^{\pi} \left(\sin(x) - 8x^2\right)\, dx$$

Solution: This problem has extremely high variance. With an estimated standard deviation of 74.74, the sample size would need to be at least 364702999. You should ask the students to estimate to within ±2, instead, which gives a more reasonable sample size of 9118.

E9 | =CONFIDENCE.T(E5,E8,E6)

	A	B	C	D	E
1				F(a) =	-1
2				F(b) =	-81.6834
3	Lower Limit a =		0	F(b) - F(a) =	-80.6834
4	Upper Limit b =		3.141592654		
5				alpha =	0.01
6				n =	9118
7				Average =	-80.7267
8				Std. Dev. =	74.165388
9				hw =	2.0010571
10				LL =	-82.72775
11				UL =	-78.72564
12					
13					
14					
15	n	U(0,1)	X ~ U(a,b)	f(X)	Y =(b-a)f(X)
16	1	0.0561	0.176121632	-0.072938105	-0.229142
17	2	0.227	0.713017199	-3.413029279	-10.72235
18	3	0.982	3.085179315	-76.09026784	-239.0446
19	4	0.443	1.391591658	-14.50823296	-45.57896
9129	9114	0.9421	2.959827539	-69.90386659	-219.6095
9130	9115	0.7712	2.422918355	-46.30587941	-145.4742
9131	9116	0.5197	1.632568094	-20.32413593	-63.85016
9132	9117	0.7713	2.422967808	-46.3078338	-145.4804
9133	9118	0.2803	0.880510561	-5.431326707	-17.06302

Question 2:

Consider the triangular distribution:

$$F(x) = \begin{cases} 0 & x < a \\ \dfrac{(x-a)^2}{(b-a)(c-a)} & a \le x \le c \\ 1 - \dfrac{(b-x)^2}{(b-a)(b-c)} & c < x \le b \\ 1 & b < x \end{cases}$$

a) Use a spreadsheet to generate 1000 observations of the triangular distribution with $a = 2$, $c = 5$, $b = 10$?

b) Use your favorite statistical software to make a histogram of 1000 observations from your implementation of the triangular distribution with $a = 2$, $c = 5$, $b = 10$?

Solution:

The triangular(a, c, b) distribution has probability density function

$$f(x) = \begin{cases} \frac{2(x-a)}{(b-a)(c-a)} & a < x < c \\ \frac{2(b-x)}{(b-a)(b-c)} & c \le x < b \end{cases}$$

and cumulative distribution function

$$F(x) = \begin{cases} \frac{(x-a)^2}{(b-a)(c-a)} & a < x < c \\ 1 - \frac{(x-b)^2}{(b-a)(b-c)} & c \le x < b. \end{cases}$$

Equating the cumulative distribution function to u, where $0 < u < 1$ yields an inverse cumulative distribution function

$$F^{-1}(u) = \begin{cases} a + \sqrt{(b-a)(c-a)u} & 0 < u < \frac{c-a}{b-a} \\ b - \sqrt{(b-a)(b-c)(1-u)} & \frac{c-a}{b-a} \le u < 1. \end{cases}$$

```
rand = Rnd()
  If rand < (mode - min) / (max - min) Then
    Triangular = min + Sqr((max - min) * (mode - min) * rand)
  Else
    Triangular = max - Sqr((max - min) * (max - mode) * (1 - rand))
  End If
End Function
```


Histogram
Frequency
0
50
100
150
200
250
2
4
6
8
10

Question 3:

A firm is trying to decide whether or not to invest in two proposals A and B that have the net cash flows shown in the following table, where $N(\mu, \sigma)$ represents that the cash flow value comes from a normal distribution with the provided mean and standard deviation.

End of Year	0	1	2	3	4
A	$N(-250, 10)$	$N(75, 10)$	$N(75, 10)$	$N(175, 20)$	$N(150, 40)$
B	$N(-250, 5)$	$N(150, 10)$	$N(150, 10)$	$N(75, 20)$	$N(75, 30)$

The interest rate has been varying recently and the firm is unsure of the rate for performing the analysis. To be safe, they have decided that the interest rate should be modeled as a beta random variable over the range from 2 to 7 percent with $\alpha_1 = 4.0$ and $\alpha_2 = 1.2$. Given all the uncertain elements in the situation, they have decided to perform a simulation analysis in order to assess the situation.

a) Compare the expected present worth of the two alternatives. Estimate the probability that alternative A has a higher present worth than alternative B.

b) Determine the number of samples needed to be 95% confidence that you have estimated the P{PW(A) > *PW(B)*} to within ± 0.10.

Solution:

a) Use the PV or NPV functions to compute the net present values. Use NORM.INV() to generate the cash flow value for each period. Use BETA.INV() to generate the interest rate. Set up two data tables to generate NPV values for each alternative. Set up statistical collection over the simulated values. Based on a sample size of 100, we can be 95% confident that the true probability is between [0.49, 0.69].

b) Based on the initial estimate of p = 0.59 for a sample size of 100, the recommended sample size to be 95% confident of being within ± 0.10 should be at least:

$$n = \left(\frac{z_{\alpha/2}}{E}\right)^2 \hat{p}(1-\hat{p}) = \left(\frac{1.96}{0.1}\right)^2 0.59(1-0.59) = 138.2$$

	A	B	C	D	E	F	G	H
1	**Inputs**							
2	min =	2%						
3	max =	7%						
4	alpha 1 =	4						
5	alpha 2 =	1.2						
6	interest rate =	5.03%						
7		period	0	1	2	3	4	
8	A	mu	-250	75	75	175	150	
9		sigma	10	10	10	20	40	
10	B	mu	-250	150	150	75	75	
11		sigma	5	10	10	20	30	
12	**Outputs**							
13	A	Series =	$253.67	$77.90	$79.24	$159.36	$143.65	
14	B	Series =	$248.02	$156.76	$153.14	$51.28	$109.51	
15								NPV's
16	A	PV =	-$253.67	$74.16	$71.82	$137.53	$118.03	$147.89
17	B	PV =	-$248.02	$149.25	$138.82	$44.25	$89.98	$174.28
18								
19	A	NPV =	$147.89					
20	B	NPV =	$174.28					

A21 fx

	A	B	C	D
1	**Inputs**			
2	min =	0.02		
3	max =	0.07		
4	alpha 1 =	4		
5	alpha 2 =	1.2		
6	interest rate =	=BETA.INV(RAND(),B4,B5,B2,B3)		
7		period	0	1
8	A	mu	-250	75
9		sigma	10	10
10	B	mu	-250	150
11		sigma	5	10
12	**Outputs**			
13	A	Series =	=NORM.INV(RAND(),C8,C9)	=NORM.INV(RAND(),D8,D9)
14	B	Series =	=NORM.INV(RAND(),C10,C11)	=NORM.INV(RAND(),D10,D11)
15				
16	A	PV =	=PV(B6,C$7,0,-C13,0)	=PV(B6,D$7,0,-D13,0)
17	B	PV =	=PV(B6,C$7,0,-C14,0)	=PV(B6,D$7,0,-D14,0)
18				
19	A	NPV =	=C16+NPV(B6,D13:G13,0)	
20	B	NPV =	=C14+NPV(B6,D14:G14,0)	

	A	B	C	D	E	F	G	H
21								
22			alpha	0.05				
23				**A**		**B**	**A-B**	**P(A>B)**
24			n =	100		100	100	100
25			average =	156.774172		146.602	10.1724	0.59
26			std dev =	38.29493627		35.179	50.4038	0.49431
27			hw =	7.59854617		6.98027	10.0012	0.09808
28			LL =	149.1756258		139.622	0.17115	0.49192
29			UL =	164.3727182		153.582	20.1736	0.68808
30								
31				$158.74		$115.48		
32			1	174.4956633	1	171.258	3.23729	1
33			2	175.7752599	2	148.451	27.3244	1
34			3	116.9952197	3	159.675	-42.68	0
35			4	141.2823175	4	200.992	-59.709	0
126			95	106.8626749	95	137.494	-30.632	0
127			96	216.127182	96	142.947	73.1805	1
128			97	151.5551783	97	146.737	4.81778	1
129			98	138.0900241	98	127.287	10.8035	1
130			99	111.9407205	99	130.904	-18.963	0
131			100	92.91871804	100	154.895	-61.976	0

Paste the contents of the clipboard

	B	C	D	E	F	G	H
22		alpha	0.05				
23			**A**		**B**	**A-B**	**P(A>B)**
24		n =	=COUNT(D32:D131)		=COUNT(F32:F131)	=COUNT(G32:G131	=COUNT(H32:H131
25		average =	=AVERAGE(D32:D131)		=AVERAGE(F32:F13	=AVERAGE(G32:G1	=AVERAGE(H32:H1
26		std dev =	=STDEV.S(D32:D131)		=STDEV.S(F32:F131	=STDEV.S(G32:G13	=STDEV.S(H32:H13
27		hw =	=CONFIDENCE.T(D22,D26,D24)		=CONFIDENCE.T($D	=CONFIDENCE.T($D	=CONFIDENCE.T($D
28		LL =	=D25-D27		=F25-F27	=G25-G27	=H25-H27
29		UL =	=D25+D27		=F25+F27	=G25+G27	=H25+H27
30							
31			=C19		=C20		
32		1	=TABLE(,C31)	1	=TABLE(,E31)	=D32-F32	=IF(G32>0,1,0)
33		2	=TABLE(,C31)	2	=TABLE(,E31)	=D33-F33	=IF(G33>0,1,0)
34		3	=TABLE(,C31)	3	=TABLE(,E31)	=D34-F34	=IF(G34>0,1,0)
35		4	=TABLE(,C31)	4	=TABLE(,E31)	=D35-F35	=IF(G35>0,1,0)
126		95	=TABLE(,C31)	95	=TABLE(,E31)	=D126-F126	=IF(G126>0,1,0)
127		96	=TABLE(,C31)	96	=TABLE(,E31)	=D127-F127	=IF(G127>0,1,0)
128		97	=TABLE(,C31)	97	=TABLE(,E31)	=D128-F128	=IF(G128>0,1,0)
129		98	=TABLE(,C31)	98	=TABLE(,E31)	=D129-F129	=IF(G129>0,1,0)
130		99	=TABLE(,C31)	99	=TABLE(,E31)	=D130-F130	=IF(G130>0,1,0)
131		100	=TABLE(,C31)	100	=TABLE(,E31)	=D131-F131	=IF(G131>0,1,0)
132							

Question 4:

Shipments can be transported by rail or trucks between New York and Los Angeles. Both modes of transport go through the city of St. Louis. The mean travel time and standard deviations between the major cities for each mode of transportation are shown in the following figure.



Assume that the travel times (in either direction) are lognormally distributed as shown in the figure. For example, the time from NY to St. Louis (or St. Louis to NY) by truck is 30 hours with a standard deviation of 6 hours. In addition, assume that the transfer time in hours in St. Louis is triangularly distributed with parameters (8, 10, 12) for trucks (truck to truck). The transfer time in hours involving rail is triangularly distributed with parameters (13, 15, 17) for rail (rail to rail, rail to truck, truck to rail). We are interested in determining the shortest total shipment time combination from NY to LA. Develop a spreadsheet simulation for this problem.

a) How many shipment combinations are there?

b) Write a spreadsheet expression for the total shipment time of the truck only combination.

c) We are interested in estimating the average shipment time for each shipment combination and the probability that the shipment combination will be able to deliver the shipment within 85 hours.

d) Estimate the probability that a shipping combination will be the shortest.

e) Determine the sample size necessary to estimate the mean shipment time for the truck only combination to within 0.5 hours with 95% confidence

Solution:

Part a is easy. Part b is reasonable. Part c is challenging. Part d is very challenging. Part e is easy if the student completed part c.

a) There are 4 shipment combinations.

b) Unfortunately, we have to solve for the normal parameters of the lognormal distribution in order to write the spreadsheet equations.

$$\mathrm{E}[X] = e^{\mu+\sigma^2/2}$$
$$\mathrm{Var}[X] = e^{2\mu+\sigma^2}\left(e^{\sigma^2}-1\right)$$

Solving for μ and σ^2

$$m = E[X]$$
$$v = V[X]$$

Then,

$$\mu = ln\left(\frac{m}{\sqrt{1+\frac{v}{m^2}}}\right)$$

$$\sigma^2 = ln\left(1+\frac{v}{m^2}\right)$$

LOGN(40,8) has mu1 = 3.669 and sigma1 = 0.198

LOGN(30,6) has mu2 = 3.3816 and sigma2 = 0.198

A single expression cannot be used to implement the triangular distribution (unless VBA is used).

X =LOGNORM.INV(RAND(),mu1, sigma1)

Y= LOGNORM.INV(RAND(),mu2, sigma2)

U = RAND()

F1 = 8 + SQRT((12-8)*(10-8)*U)

F2 = 12 – SQRT((12-8)*(12-10)*(1-U))

Z = If (U< (10-8)/(12-8), F1, F2)

Time = X + Y + Z

c) You should notice that truck takes longer than rail, which is not realistic.

Outline:

1) set up to compute the mu and sigma for the lognormal distributions, see part b
2) set up to generate from the lognormal distributions, see part b

3) set up to generate from triangular distribution. See book appendix for CDF formulas and part b
4) enumerate the 4 path combinations and their lengths
5) use data tables to generate path lengths
6) use indicator variables to estimate P{path length <=85}
7) use MIN() function to find shortest path length
8) use IF() function to indicate which was the shortest. Collect statistics on indicators.

N24

	A	B	C	D	E	F
1	Inputs	LA to STL by T	STL to NY by T	LA to STL by	STL to NY by R	
2	mean =	40	30	36	24	
3	std dev =	8	6	5.4	3.60	
4	v/m^2 =	0.04	0.04	0.0225	0.0225	
5	mu =	3.6692691	3.3815870	3.5723936	3.1669285	
6	sigma =	0.1980422	0.1980422	0.1491664	0.1491664	
7	Time	53.51664135	29.6468677	33.74056	32.0306738	
8						
9	Transfer Time	T to T	R to R	R to T	T to R	
10	min =	8	13	13	13	
11	mode =	10	15	15	15	
12	max =	12	17	17	17	
13	u =	0.171937178	0.37099371	0.371246	0.28155558	
14	Finv 1 =	9.172816023	14.7227738	14.72336	14.5008147	
15	Finv 2 =	9.426189095	14.7567768	14.75723	14.602594	
16	(m-a)/(b-a) =	0.5	0.5	0.5	0.5	
17	Finv =	9.172816023	14.7227738	14.72336	14.5008147	
18						
19	Combinations				Total	
20	All Truck	53.51664135	9.17281602	29.64687	92.3363251	
21	All Rail	33.74056432	14.7227738	32.03067	80.4940119	
22	First Truck then Rail	53.51664135	14.5008147	32.03067	100.04813	
23	First Rail then Truck	33.74056432	14.7233587	29.64687	78.1107907	

	A	B	C	D	E
1	Inputs	LA to STL by T	STL to NY by T	LA to STL by R	STL to NY by R
2	mean =	40	30	36	24
3	std dev =	8	6	5.4	3.6
4	v/m^2 =	=(B3*B3)/(B2*B2)	=(C3*C3)/(C2*C2)	=(D3*D3)/(D2*D2)	=(E3*E3)/(E2*E2)
5	mu =	=LN(B2/SQRT(1+B4))	=LN(C2/SQRT(1+C4))	=LN(D2/SQRT(1+D4))	=LN(E2/SQRT(1+E4))
6	sigma =	=SQRT(LN(1+B4))	=SQRT(LN(1+C4))	=SQRT(LN(1+D4))	=SQRT(LN(1+E4))
7	Time	=LOGNORM.INV(RAND(),B5,B6)	=LOGNORM.INV(RAND(),C5,C	=LOGNORM.INV(RAND(	=LOGNORM.INV(RAND(),E5,E
8					
9	Transfer Time	T to T	R to R	R to T	T to R
10	min =	8	13	13	13
11	mode =	10	15	15	15
12	max =	12	17	17	17
13	u =	=RAND()	=RAND()	=RAND()	=RAND()
14	Finv 1 =	=B10+SQRT((B12-B10)*(B11-B10)*B13)	=C10+SQRT((C12-C10)*(C11-C	=D10+SQRT((D12-D10)*	=E10+SQRT((E12-E10)*(E11-E
15	Finv 2 =	=B12-SQRT((B12-B10)*(B11-B10)*(1-B13))	=C12-SQRT((C12-C10)*(C11-C	=D12-SQRT((D12-D10)*	=E12-SQRT((E12-E10)*(E11-E
16	(m-a)/(b-a) =	=(B11-B10)/(B12-B10)	=(C11-C10)/(C12-C10)	=(D11-D10)/(D12-D10)	=(E11-E10)/(E12-E10)
17	Finv =	=IF(B13<B16,B14,B15)	=IF(C13<C16,C14,C15)	=IF(D13<D16,D14,D15)	=IF(E13<E16,E14,E15)
18					
19	Combinations				Total
20	All Truck	=B7	=B17	=C7	=SUM(B20:D20)
21	All Rail	=D7	=C17	=E7	=SUM(B21:D21)
22	First Truck then Rail	=B7	=E17	=E7	=SUM(B22:D22)
23	First Rail then Truck	=D7	=D17	=C7	=SUM(B23:D23)

	A	B	C	D	E	F	G	H	I	J	K	L
24												
25		All Truck	P{X<=85}		All Rail	P{X<=85}		Truck then Rail	P{X<=85}		Rail then Truck	P{X<=85}
26	alpha =	0.05	0.05	alpha =	0.05	0.05	alpha =	0.05	0.05	alpha =	0.05	0.05
27	n =	100	100	n =	100	100	n =	100	100	n =	100	100
28	avg =	79.52527176	0.73	avg =	74.68637671	0.93	avg =	80.03102447	0.74	avg =	80.56256193	0.78
29	s =	9.010727484	0.446196043	s =	6.184982925	0.25643	s =	8.434673756	0.440844	s =	7.506884871	0.4163332
30	hw =	1.787923822	0.088534975	hw =	1.227234796	0.05088	hw =	1.673622265	0.08747301	hw =	1.489528821	0.08260954
31	LL	77.73734794	0.641465025	LL	73.45914191	0.87912	LL	78.35740221	0.65252699	LL	79.0730331	0.69739046
32	UL	81.31319558	0.818534975	UL	75.91361151	0.98088	UL	81.70464674	0.82747301	UL	82.05209075	0.86260954
33												
34		All Truck			All Rail			Truck then Rail			Rail then Truck	
35		72.06941966			64.33554821			76.75365666			69.21876547	
36	1	73.25039613	1	1	75.65630363	1	1	100.6877029	0	1	73.78028432	1
37	2	66.48900663	1	2	77.90429637	1	2	70.12358724	1	2	81.04543148	1
38	3	78.7096026	1	3	78.61842064	1	3	75.33992129	1	3	85.11109523	0
39	4	94.49751438	0	4	70.97492491	1	4	79.99842781	1	4	88.86370209	0
40	5	76.81598717	1	5	86.19457552	0	5	88.05566579	0	5	81.93337936	1
130	95	86.73050939	0	95	71.05648267	1	95	84.95508096	1	95	80.30205056	1
131	96	80.03143255	1	96	79.18214796	1	96	82.56772806	1	96	73.76312773	1
132	97	85.09739234	0	97	73.16613769	1	97	75.25118535	1	97	83.33916538	1
133	98	85.01466525	0	98	73.62575525	1	98	86.01217741	0	98	78.22297361	1
134	99	95.59705123	0	99	75.1286691	1	99	85.62326658	0	99	59.75365043	1
135	100	84.79726922	1	100	79.09351831	1	100	71.12382279	1	100	92.7507277	0

d)

	N	O	P	Q	R
25					
26	0.05	0.05	0.05	0.05	0.05
27	100	100	100	100	100
28	70.657906	0.25	0.41	0.17	0.17
29	5.1842601	0.43519	0.494	0.377525168	0.377525168
30	1.0286697	0.08635	0.098	0.074909184	0.074909184
31	69.629236	0.16365	0.312	0.095090816	0.095090816
32	71.686576	0.33635	0.508	0.244909184	0.244909184
33					
34					
35	Min	All Truck	All Rail	Truck then Rail	Rail then Truck
36	73.250396	1	0	0	0
37	66.489007	1	0	0	0
38	75.339921	0	0	1	0
39	70.974925	0	1	0	0
40	76.815987	1	0	0	0
130	71.056483	0	1	0	0
131	73.763128	0	0	0	1
132	73.166138	0	1	0	0
133	73.625755	0	1	0	0
134	59.75365	0	0	0	1
135	71.123823	0	0	1	0

e) Using the results from part (c)

$$n \geq \left(\frac{z_{\alpha/2}s}{E}\right)^2 = \left(\frac{1.96 \times 9.01}{0.5}\right)^2 = 1247.446 = 1248$$

Question 5:

See Question 1 in solutions of Exercises Sheet #5.

Question 6:

A firm produces YBox gaming stations for the consumer market. Their profit function is:

Profit = (unit price - unit cost) × (quantity sold) − fixed costs

Suppose that the unit price is \$200 per gaming station, and that the other variables have the following probability distributions:

Unit Cost	80	90	100	110
Probability	0.20	0.40	0.30	0.10

Quantity Sold	1000	2000	3000
Probability	0.10	0.60	0.30

Fixed Cost	50000	65000	80000
Probability	0.40	0.30	0.30

Use a spreadsheet to generate 1000 observations of the profit.

a) Make a histogram of your observations using your favorite statistical analysis package.

b) Estimate the mean profit from your sample and compute a 99% confidence interval for the mean profit.

c) Estimate the probability that the profit will be positive.

Solution:

a)



b)

	A	B	C	D	E	F	G
1	Inputs					Unit Price =	$200.00
2						Unit Cost =	$100.00
3	p(x)	LL	F(x)	Unit Cost		Qty Sold =	2000
4	0.2	0	0.2	80		Fixed Cost =	$65,000.00
5	0.4	0.2	0.6	90		Profit =	$135,000.00
6	0.3	0.6	0.9	100			
7	0.1	0.9	1	110			
8							
9	p(x)	LL	F(x)	Qty Sold	alpha =	0.01	
10	0.1	0	0.1	1000	n =	1000	
11	0.6	0.1	0.7	2000	avg =	$168,665.00	
12	0.3	0.7	1	3000	s =	$69,193.07	
13					hw =	$5,646.90	
14	p(x)	LL	F(x)	Fixed Cos	LL =	$163,018.10	
15	0.4	0	0.4	50000	UL =	$174,311.90	
16	0.3	0.2	0.7	65000	min =	10000	
17	0.3	0.6	1	80000	max =	310000	
18							
19						$135,000.00	
20					1	160000.00	
21					2	170000.00	
22					3	30000.00	

c) The estimate is 1. All profits were positive in the sample

Question 7:

T. Wilson operates a sports magazine stand before each game. He can buy each magazine for 33 cents and can sell each magazine for 50 cents. Magazines not sold at the end of the game are sold for scrap for 5 cents each. Magazines can only be purchased in bundles of 10. Thus, he can buy 10, 20, and so on magazines prior to the game to stock his stand. The lost revenue for not meeting demand is 17 cents for each magazine demanded that could not be provided. Mr. Wilson's profit is as follows:

$$\begin{aligned}\text{Profit} = &(\text{revenue from sales}) - (\text{cost of magazines})\\ &- (\text{lost profit from excess demand})\\ &+ (\text{salvage value from sale of scrap magazines})\end{aligned}$$

Not all game days are the same in terms of potential demand. The type of day depends on a number of factors including the current standings, the opponent, and whether or not there are other special events planned for the game day weekend. There are three types of game days demand: high, medium, low. The type of day has a probability distribution associated with it.

Type of Day	High	Medium	Low
Probability	0.35	0.45	0.20

The amount of demand for magazines then depends on the type of day according to the following distributions:

	High		Medium		Low	
Demand	PMF	CDF	PMF	CDF	PMF	CDF
40	0.03	0.03	0.1	0.1	0.44	0.44
50	0.05	0.08	0.18	0.28	0.22	0.66
60	0.15	0.23	0.4	0.68	0.16	0.82
70	0.2	0.43	0.2	0.88	0.12	0.94
80	0.35	0.78	0.08	0.96	0.06	1.0
90	0.15	0.93	0.04	1.0		
100	0.07	1.0				

Solution:

Let Q be the number of units of magazines purchased (quantity on hand) to setup the stand. Let D represent the demand for the game day. If D > Q, Mr. Wilson sells only Q and will have lost sales of D-Q. If D < Q, Mr. Wilson sells only D and will have scrap of Q-D. Assume that he has determined that Q = 50.

Make sure that you can estimate the average profit and the probability that the profit is greater than zero for Mr. Wilson. Develop a spreadsheet model to estimate the average profit with 95% confidence to within plus or minus \$0.5.

Initial sample size of 100 yields, s = 13.4. Thus, sample size needed is:

$$n \geq \left(\frac{z_{\alpha/2}s}{E}\right)^2 = \left(\frac{1.96 \times 13.4}{0.5}\right)^2 = 2759.19 = 2760$$

The order policy of Q = 50 is profitable on average and there is 97% chance of positive profits with this policy. You should ask the students to try to find an optimal policy for this situation.

	A	B	C	D	E	F	G	H	I	J
1				**Type of Day**				u =	0.85543321	
2	**Inputs**			**PMF**	**LR**	**CDF**	**D**	Day =	3	
3	cost per unit c =	0.33		0.35	0	0.35	1			
4	price per unit s =	0.5		0.45	0.35	0.8	2			
5	salvage per unit u =	0.05		0.2	0.8	1	3			
6	lost sales per unit =	0.17						u =	0.66538301	
7				**High Demand Distribution**						
8	**Decision Variable**			**PMF**	**LR**	**CDF**	**D**	High Demand	80	
9	Order Quantiy q =	50		0.03	0	0.03	40			
10				0.05	0.03	0.08	50			
11	**Outputs**			0.15	0.08	0.23	60			
12	Daily Demand	60		0.2	0.23	0.43	70			
13	Amount Sold	50		0.35	0.43	0.78	80			
14	Lost Sales	10		0.15	0.78	0.93	90			
15	Amount Left Over	0		0.07	0.93	1	100			
16	Sales Revenue	25								
17	Salvage Revenue	0		**Medium Demand Distribution**						
18	Ordering Cost	16.5		**PMF**	**LR**	**CDF**	**D**	Medium Demand	60	
19	Lost Sales Cost	1.7		0.1	0	0.1	40			
20	Profit	6.8		0.18	0.1	0.28	50			
21				0.4	0.28	0.68	60			
22	**Simulation Summary**			0.2	0.68	0.88	70			
23	alpha =	0.05	0.05	0.08	0.88	0.96	80			
24	n =	2760	2760	0.04	0.96	1	90			
25	Average =	5.20986	0.9739							
26	Std Dev =	2.14063	0.1594	**Low Demand Distribution**						
27	hw =	0.0799	0.006	**PMF**	**LR**	**CDF**	**D**	Low Demand	60	
28	LL =	5.12996	0.968	0.44	0	0.44	40			
29	UL =	5.28975	0.9799	0.22	0.44	0.66	50			
30	min =	0		0.16	0.66	0.82	60			
31	max =	8.5		0.12	0.82	0.94	70			
32				0.06	0.94	1	80			
33	**Data Table**									
34		6.8	1							
35	1	3.4	1							
126	92	6.8	1							
127	93	3.4	1							
128	94	8.5	1							
129	95	6.8	1							

P3-29 | Sheet1 | Normal View | Ready | Sum=0.22

Question 8:

The time for an automated storage and retrieval system in a warehouse to locate a part consists of three movements. Let X be the time to travel to the correct aisle. Let Y be the time to travel to the correct location along the aisle. And let Z be the time to travel up to the correct location on the shelves. Assume that the distributions of X, Y, and Z are as follows:

- $X \sim$ lognormal with mean 20 and standard deviation 10 seconds
- $Y \sim$ uniform with minimum 10 and maximum 15 seconds
- $Z \sim$ uniform with minimum of 5 and a maximum of 10 seconds

Develop a spreadsheet that can estimate the average total time that it takes to locate a part and can estimate the probability that the time to locate a part exceeds 60 seconds. Base your analysis on 1000 observations.

Solution:

	A	B	C	D	E	F	G	H
1	Inputs	X						
2	mean =	20						
3	std dev =	10						
4	v/m^2 =	0.25						
5	mu =	2.8841605						
6	sigma =	0.4723807						
7	Time	6.862490838						
8								
9	Transfer Time	Y	Z					
10	min =	10	5					
11	max =	15	10					
12	u =	0.261316855	0.13505909					
13	Finv =	11.30658427	5.67529547					
14								
15	Total =	23.84437058						
16								
17		Total	P{Total<=60}					
18	alpha =	0.05	0.05					
19	n =	1000	1000					
20	avg =	40.27470217	0.953					
21	s =	10.22987036	0.21174474					
22	hw =	0.634811389	0.01313975					
23	LL	39.63989078	0.93986025					
24	UL	40.90951356	0.96613975					
25								
26		Total						
27		23.84437058						
28	1	47.28434278	1					
29	2	35.49313259	1					
30	3	37.73234166	1					

P3-30

Normal View   Ready   Sum=0

Question 9:

Lead-time demand may occur in an inventory system when the lead-time is other than instantaneous. The lead-time is the time from the placement of an order until the order is received. The lead-time is a random variable. During the lead-time, demand also occurs at random. Lead-time demand is thus a random variable defined as the sum of the demands during the lead-time, or LDT = $\sum_{i=1}^{T} D_i$ where i is the time period of the lead-time and T is the lead-time. The distribution of lead-time demand is determined by simulating many cycles of lead-time and the demands that occur during the lead-time to get many realizations of the random variable LDT. Notice that LDT is the *convolution* of a random number of random demands. Suppose that the daily demand for an item is given by the following probability mass function:

Daily Demand (items)	4	5	6	7	8
Probability	0.10	0.30	0.35	0.10	0.15

The lead-time is the number of days from placing an order until the firm receives the order from the supplier.

a) Assume that the lead-time is a constant 10 days. Use a spreadsheet to simulate 1000 instances of LDT. Report the summary statistics for the 1000 observations. Estimate the chance that LDT is greater than or equal to 10. Report a 95% confidence interval on your estimate. Use your favorite statistical program to develop a frequency diagram for LDT.

b) Assume that the lead-time has a shifted geometric distribution with probability parameter equal to 0.2 Use a spreadsheet to simulate 1000 instances of LDT. Report the summary statistics for the 1000 observations. Estimate the chance that LDT is greater than or equal to 10. Report a 95% confidence interval on your estimate. Use your favorite statistical program to develop a frequency diagram for LDT.

Solution:

a) Because the lead time is constant, the solution involves generating 10 demands and summing them up. Then use a data table to generate 1000 of these sums. The chance that the LDT is greater than 10 is 1.0.

	A	B	C	D
1	**Demand Distribution**			
2	**PMF**	**LR**	**CDF**	**D**
3	0.1	0	0.1	4
4	0.3	0.1	0.4	5
5	0.35	0.4	0.75	6
6	0.1	0.75	0.85	7
7	0.15	0.85	1	8
8				
9	Day	Demand		
10	1	8		
11	2	7		
12	3	5		
13	4	5		
14	5	6		
15	6	5		
16	7	4		
17	8	7		
18	9	7		
19	10	6		
20	LTD =	60		
21	**Summary**			
22	alpha =	0.05	0.05	
23	n =	1000	1000	
24	Average =	58.895	1	
25	Std Dev =	3.70725	0	
26	hw =	0.23005	#NUM!	
27	LL =	58.6649	#NUM!	
28	UL =	59.1251	#NUM!	

b) This part is more difficult to do in a spreadsheet because of the random sum that must be generated because of the shifted geometric distribution. To generate from a shifted geometric, use the following:

=1+INT(LN(1-RAND())/LN(1-p))

A simple solution is to create a running sum of the demand and to use a VLOOKUP() to return the sum based on the value of the lead time. The only technical issue is that the column that holds the running sum needs to be long (theoretically infinite) because the geometric distribution has infinite range. From a practical standpoint, just make it long enough to ensure with high probability that there are enough sums to look up. Alternatively, a VBA function can be used.

	A	B	C	D	E	F	G	H
1	**Demand Distribution**							
2	**PMF**	**LR**	**CDF**	**D**		Day	Demand	Sum
3	0.1	0	0.1	4		1	6	6
4	0.3	0.1	0.4	5		2	4	10
5	0.35	0.4	0.75	6		3	4	14
6	0.1	0.75	0.85	7		4	7	21
7	0.15	0.85	1	8		5	6	27
8						6	6	33
9	p=	0.2				7	5	38
10	Lead Time	10				8	5	43
11	LTD =	55				9	6	49
12	**Summary**					10	6	55
13	alpha =	0.05	0.05			11	7	62
14	n =	1000	1000			12	6	68
15	Average =	29.115	0.757			13	5	73
16	Std Dev =	25.9044	0.42911			14	5	78
17	hw =	1.60749	0.026628			15	7	85
18	LL =	27.5075	0.730372			16	4	89
19	UL =	30.7225	0.783628			17	6	95
20	min =	4				18	5	100
21	max =	207				19	6	106
22						20	8	114

Question 10:

See Question 2 in Solutions of Exercises Sheet #5.

Question 11:

See Question 7 in Solutions of Exercises Sheet #5.

Question 12:

Describe a simulation experiment that would allow you to test the lack of memory property empirically. Implement your simulation experiment in a spreadsheet and test the lack of memory property empirically. Explain in your own words what lack of memory means.

Solution:

Suppose the mean is 100 and delta is 120 and t1=200 and therefore t2 = 320. The key is to record statistics only on those cases where T > 200 AND T>320. You can then compare this to only those where T>120.

```
For I = 1 to n
        Let T ~ expo(100)
        If T > 120 record 1, else record 0 as P(T>120)
        If T > 200
                If T > 320, record 1, else record 0 as P(T>320| T > 200)
        End if
End for
```

	A	B	C	D	E	F
1	mean =	100		True =	0.30119421	
2	delta =	120				
3	t1 =	200				
4	t2 =	320				0.280303
5			0.2875	264		74
6	n	T	P(T>delta)	T > t1	T>t2|T>t1	
7	1	100.93992	0	FALSE	FALSE	0
8	2	199.23572	1	FALSE	FALSE	0
9	3	6.7276793	0	FALSE	FALSE	0
10	4	285.93253	1	TRUE	FALSE	0
11	5	24.504143	0	FALSE	FALSE	0
2001	1995	160.94289	1	FALSE	FALSE	0
2002	1996	51.278586	0	FALSE	FALSE	0
2003	1997	78.256211	0	FALSE	FALSE	0
2004	1998	111.97074	0	FALSE	FALSE	0
2005	1999	134.74332	1	FALSE	FALSE	0
2006	2000	279.32884	1	TRUE	FALSE	0
2007						

	A	B	C	D	E	F
1	mean =	100		True =	=EXP(-B2/B1)	
2	delta =	120				
3	t1 =	200				
4	t2 =	=B3+B2				=F5/D5
5			=AVERAGE(C7:C2006)	=COUNTIF(D7:D2006,TRUE)		=SUM(F7:F2006)
6	n	T	P(T>delta)	T > t1	T>t2|T>t1	
7	1	=-B1*LN(1-RAND())	=IF(B7>B2,1,0)	=IF(B7>B3,TRUE, FALSE)	=IF(B7>B4,TRUE,FALSE)	=IF(AND(D7,E7),1,0)
8	=A7+1	=-B1*LN(1-RAND())	=IF(B8>B2,1,0)	=IF(B8>B3,TRUE, FALSE)	=IF(B8>B4,TRUE,FALSE)	=IF(AND(D8,E8),1,0)
9	=A8+1	=-B1*LN(1-RAND())	=IF(B9>B2,1,0)	=IF(B9>B3,TRUE, FALSE)	=IF(B9>B4,TRUE,FALSE)	=IF(AND(D9,E9),1,0)
10	=A9+1	=-B1*LN(1-RAND())	=IF(B10>B2,1,0)	=IF(B10>B3,TRUE, FALSE)	=IF(B10>B4,TRUE,FALSE)	=IF(AND(D10,E10),1,0)
11	=A10+1	=-B1*LN(1-RAND())	=IF(B11>B2,1,0)	=IF(B11>B3,TRUE, FALSE)	=IF(B11>B4,TRUE,FALSE)	=IF(AND(D11,E11),1,0)
2001	=A2000+1	=-B1*LN(1-RAND())	=IF(B2001>B2,1,0)	=IF(B2001>B3,TRUE, FALSE)	=IF(B2001>B4,TRUE,FALSE)	=IF(AND(D2001,E2001),1,0)
2002	=A2001+1	=-B1*LN(1-RAND())	=IF(B2002>B2,1,0)	=IF(B2002>B3,TRUE, FALSE)	=IF(B2002>B4,TRUE,FALSE)	=IF(AND(D2002,E2002),1,0)
2003	=A2002+1	=-B1*LN(1-RAND())	=IF(B2003>B2,1,0)	=IF(B2003>B3,TRUE, FALSE)	=IF(B2003>B4,TRUE,FALSE)	=IF(AND(D2003,E2003),1,0)
2004	=A2003+1	=-B1*LN(1-RAND())	=IF(B2004>B2,1,0)	=IF(B2004>B3,TRUE, FALSE)	=IF(B2004>B4,TRUE,FALSE)	=IF(AND(D2004,E2004),1,0)
2005	=A2004+1	=-B1*LN(1-RAND())	=IF(B2005>B2,1,0)	=IF(B2005>B3,TRUE, FALSE)	=IF(B2005>B4,TRUE,FALSE)	=IF(AND(D2005,E2005),1,0)
2006	=A2005+1	=-B1*LN(1-RAND())	=IF(B2006>B2,1,0)	=IF(B2006>B3,TRUE, FALSE)	=IF(B2006>B4,TRUE,FALSE)	=IF(AND(D2006,E2006),1,0)
2007						

Think of the random variable as the life time of a component. The memory-less property implies that the component is just as likely to survive for at least 120 more time units, when it has already survived 200 time units as it was to survive for at least 120 time units when it was new. Thus, it "forgets" the past 200 time units.

Question 13:

See Question 9 in Solutions of Exercises Sheet #5.